

# Object Action Language™ Manual

*Document Version: 1.4*

# Object Action Language™ Manual

---

**BridgePoint®** is a registered trademark of Project Technology, Inc. and its licensors. Object Action Language is a trademark of Project Technology, Inc. UML is a trademark of The Object Management Group.

All other products or services mentioned in this document are identified by the trademark, service marks, or products names as designated by the companies who market these products.

---

## **Corporate Headquarters**

Project Technology, Inc.  
7400 North Oracle Road, Suite 365  
Tucson, AZ 85704-6342  
United States

toll-free: (800) 845-1489  
voice: +1 (520) 544-2881  
fax: +1 (520) 544-2912  
email: [info@projtech.com](mailto:info@projtech.com)  
web: [www.projtech.com](http://www.projtech.com)

## **Support**

email: [support@projtech.com](mailto:support@projtech.com)  
toll-free: (800) 482-3853  
voice: +1 (520) 544-0808  
fax: +1 (520) 544-2912  
web: [www.projtech.com/support](http://www.projtech.com/support)

---

# CONTENTS

---

<b>1</b>	<b>Overview .....</b>	<b>1</b>
1.1	Documentation Road Map .....	1
1.2	Typographical Conventions.....	1
1.3	Contacting Project Technology .....	2
<b>2</b>	<b>Introduction.....</b>	<b>3</b>
2.1	Purpose.....	3
2.2	Basic Concepts.....	4
2.3	Intended Audience .....	5
2.4	Additional Conventions.....	5
2.5	Examples.....	6
<b>3</b>	<b>Language Structure .....</b>	<b>7</b>
3.1	Overall Structure .....	7
3.2	Comments .....	7
3.3	Names and Keywords.....	7
3.4	White Space .....	8

<b>4</b>	<b>Data Items.....</b>	<b>9</b>
4.1	Data Items Within an Action.....	9
4.2	Modeled Elements.....	10
4.3	Local Variables.....	11
4.4	Assigning Data Types .....	12
4.5	Variable Initialization.....	13
4.6	Scoping.....	14
<b>5</b>	<b>Control Structures .....</b>	<b>17</b>
5.1	If Construct .....	17
5.2	For Each Loop.....	19
5.3	While Loop .....	20
5.4	Break .....	21
5.5	Continue .....	22
5.6	Nested Control Logic .....	23
<b>6</b>	<b>Class Manipulations.....</b>	<b>25</b>
6.1	Creating Instances .....	25
6.2	Selecting Instances.....	26
6.3	Writing Attributes .....	28
6.4	Writing Mathematically-Dependent Attributes.....	29
6.5	Reading Attributes .....	31
6.6	Deleting Instances.....	32

---

<b>7</b>	<b>Relationships .....</b>	<b>35</b>
	7.1 Relationship Specifications .....	35
	7.2 Creating an Instance of a Relationship .....	36
	7.3 Deleting an Instance of a Relationship .....	38
	7.4 Relationship Navigation .....	40
<b>8</b>	<b>Events .....</b>	<b>43</b>
	8.1 Receiving Event Data .....	43
	8.2 Event Generation .....	44
	8.3 Event Pre-creation .....	46
	8.4 Sending a Pre-created Event .....	47
<b>9</b>	<b>Expressions .....</b>	<b>49</b>
	9.1 Simple Expressions .....	49
	9.2 Compound Expressions .....	50
	9.3 Arithmetic Expressions .....	52
	9.4 Boolean Expressions .....	53
	9.5 String Expressions .....	55
	9.6 Where Expressions .....	55
	9.7 Assignment of Variables .....	56
	9.8 Constants .....	57
	9.9 Additional Unary Operators .....	58

<b>10</b>	<b>Operations, Bridges, and Functions .....</b>	<b>61</b>
	10.1 Operation Invocation.....	61
	10.2 Bridge Invocation .....	63
	10.3 Function Invocation.....	66
	10.4 Object Action Language for Invocations .....	68
<b>11</b>	<b>Date and Time .....</b>	<b>71</b>
	11.1 External and Internal Time.....	71
<b>12</b>	<b>Timers .....</b>	<b>73</b>
	12.1 Starting a Timer.....	73
	12.2 Querying a Timer .....	74
	12.3 Manipulating a Timer.....	75
	12.4 Canceling a Timer .....	76
<b>A</b>	<b>References .....</b>	<b>77</b>
	A.1 References.....	77
<b>B</b>	<b>Keywords .....</b>	<b>79</b>
	B.1 Keywords .....	79
<b>C</b>	<b>Syntax Summary .....</b>	<b>81</b>
	C.1 Language Constructs.....	81
	C.2 Statement Components .....	88
	C.3 Production Rules .....	89
<b>D</b>	<b>Index.....</b>	<b>95</b>

---

## 1.1 Documentation Road Map

### 1.1.1 Object Action Language Manual

The *Object Action Language Manual* details an action language that realizes the action semantics specification for UML as of version 1.5. Please check the OMG website for the latest UML specification.

The Object Action Language described here is fully supported by the BridgePoint Development Suite.

## 1.2 Typographical Conventions

The following typographical conventions are used throughout this document. A different font is used to bring attention to different textual elements:

<code>command</code>	Denotes a command as it should be typed in, a file name, or a class name.
<code>&lt;value&gt;</code>	Denotes a value supplied by the user including command-line parameters or directory paths.
<i>Start / Programs</i>	Indicates a sequence of menu selections or series of buttons to be selected under the windowing environment.
<i>Document</i>	Refers to documents outside the current one.

See “Section”

Refers to a section or subsection in the current document.

## 1.3 **Contacting Project Technology**

The following lines of communication can be used to contact Project Technology:

	<i>Phone</i>	<i>Fax</i>	<i>E-mail</i>
<b>PT Sales</b>	+1 (520) 544-2881 (800) 845-1489	+1 (520) 544-2912	sales@projtech.com
<b>PT Support</b>	+1 (520) 544-0808 (800) 482-3853	+1 (520) 544-2912	support@projtech.com

You can also visit our web site at [www.projtech.com](http://www.projtech.com).



---

## **2.1 Purpose**

The purpose of this manual is to serve as a reference and general user's guide to aid in the correct specification of action semantics for UML models. Although originally designed for models used with the BridgePoint Development Suite, the language described herein can be used to define the action semantics for any UML model in any tool.

The Object Action Language™ is written to satisfy the following goals:

- Readability - Modelers must be able to easily understand the OAL for development and reviews.
- Derivation - Event generation and data access information is captured for derivation of the Object Collaboration Diagrams and Package Dependency Diagrams for both asynchronous (event) and synchronous (data access) communication.
- Simulation - The UML models can be simulated through interpretation of the actions by using the Model Verifier tool.
- Translation - Richness of expression is provided while maintaining a specification that can be automatically translated onto a target architecture.

## 2.2 Basic Concepts

The Object Action Language™ (OAL) is used to define the semantics for the processing that occurs in an action. An action can be associated with the following five modeled elements:

- states
- bridge operations
- functions
- class and instance-based operations
- mathematically-dependent attributes

The Object Action Language provides for five types of action processes:

- data access
- event generation
- test
- transformation
- bridge and function

It supports these through:

- control logic
- access to the data described by the class diagram
- access to the data supplied by events initiating actions
- the ability to generate events
- access to timers and to the current time and date

In a UML model, unlike conventional programming, there is no concept of a "main" function or routine where execution starts. Rather, the models are executed in the context of a number of interacting finite state machines, all of which are considered to be executing concurrently. Any state machine, upon receipt of an event (from another state machine or from outside the system) may respond by changing state. On entry to the new state, a block of processing (an

"action") is performed. This processing can in principle execute at the same time as processing associated with another state machine. (Whether this occurs in practice depends on the nature of the software and hardware architectures used to implement the system.)

The Object Action Language is used to define the processing executed during the action. The execution rules are as follows:

- Execution commences at the first statement in the action and proceeds sequentially through the succeeding lines as directed by any control logic structures.
- Execution of the action terminates when the last statement is completed.

These rules also apply to actions defined for bridge operations, functions, class and instance-based operations, and mathematically-dependent attributes.

## 2.3 *Intended Audience*

This manual is written for modelers and software engineers who use a process and set of modeling tools that support the creation, simulation, and translation of UML models. Specifically, guidance is provided for the correct specification of actions formulated in the syntax of the Object Action Language.

The following section provides guidance on using this manual to meet these needs.

## 2.4 *Additional Conventions*

Although future versions of the most tools will support models created under previous releases, this manual is annotated with three special symbols to indicate preferred usage in the light of future development.

- ☺ Indicates a preferred construct or recommended use for a given construct.
- ☹ Indicates a feature that should be used only in the limited manner described in this manual. If the feature is used for other purposes, it is unlikely to convert properly in future releases or translate correctly onto some architectures.
- ⊗ Marks a form or capability that is included in this release only for experimental reasons or for backward compatibility. The capability may not be supported in future releases of tools.

## 2.5 Examples

There are examples presented throughout this manual. Much of the OAL in these examples was extracted from a complete model of a real-world device (an auto-sampler used for chemical testing). The complete model is available on the Project Technology web site:

[www.projtech.com/prods/bp/autosamp.html](http://www.projtech.com/prods/bp/autosamp.html)

This model is intentionally simplistic in nature to make it easy to understand. Its purpose is to illustrate as many aspects of the OAL constructs as possible. It should, therefore, not be construed an example of good modeling practices as several compromises have been made in the interest of simplicity.

In many cases the examples from the auto-sampler model are augmented with additional, contrived examples in an effort to provide further illustrations of the OAL construct in question. As with any contrived examples, these sequences of OAL are provided as examples of the syntax for the language and not as examples of valid modeling techniques for real world applications.

# LANGUAGE STRUCTURE

---

## 3.1 Overall Structure

An action consists of a number of statements. Each statement can be either a simple statement (such as an access to the attributes of a class) or a control logic structure (such as an if construct).

OAL statements are terminated by a semi-colon except following the if, for each, and while control constructs, described later.

## 3.2 Comments

Comments may be inserted by the use of the // characters at any point in the line. When this pair of characters is detected, the remainder of the line (up to the new-line character) is considered to be a comment and is ignored.

## 3.3 Names and Keywords

OAL statements are composed of:

- keywords (See "Appendix B: Keywords".)
- logical and arithmetic operations
- names of modeled elements (classes, attributes, class and external entity keyletters, relationship numbers and phrases, event labels and meanings, and supplemental data items)
- local variables

Keywords may be represented in all upper case, all lower case, or with the first character upper case and all other characters in lower case.

Names in OAL statements must conform to the following rules:

1. Names are case sensitive. This includes local variables, keyletters, class attributes, and event identifiers.
2. Names must not begin with a numeric character [0-9].
3. A relationship phrase ('is owned by') or event meaning ('turn off pump')
  - may contain any ASCII characters.
  - must be enclosed by tick marks.<sup>1</sup>
  - must be contained on a single line.
4. Class keyletters, event labels, and attribute names may contain only the characters [a-z][A-Z][0-9][\_#]. Spaces are not permitted.

### **3.4 White Space**

White space (spaces and tabs) may be inserted at any point in an OAL statement other than:

- within a name or keyword,
- between two consecutive colons, or
- between the characters of a comparison operator.

---

1. The tick marks may be omitted from an event meaning if it contains no spaces.

---

### **4.1 *Data Items Within an Action***

The OAL expression of an action has access to and can produce certain data items. The following data items are available to be read at the start of and throughout an action:

- constants
- values of attributes of classes
- supplemental data items carried by the event that initiated the action
- local variables (created by statements within the action)

These data items can be produced during an action:

- local variables
- values of attributes of classes
- supplemental data items to be carried by an event generated during the action

Finally, the instance handle `self` may be used to refer to the currently executing instance in an instance statechart (but not in a class statechart), to an instance in an instance-based operation, and to the instance of a mathematically-dependent attribute.

## 4.2 Modeled Elements

All data items referenced or produced by an action must have a data type. The following data types are defined for class attributes, supplemental data items of an event, and local variables.

integer	string	unique ID	☹ timer handle
real	date	instance handle	☹ event instance
boolean	timestamp	instance handle set	☹ state <sup>a</sup>

**Table 4-1:** Object Action Language Data Types

- a. In the *OOA-96 Report*, the current state attribute is entirely under the control of the architectural domain and is not available to the analyst; as a result, in future versions this data type will no longer be needed.

The analyst may also define domain-specific data types based upon these data types.

Data types within the OAL are "analysis" data types, and reflect only the set of legal values a variable can take on.

### 4.2.1 Implicit Typing

There are no type declaration statements in the OAL. All data items are implicitly typed by the value assigned to them on their first use within an action.



## 4.3 Local Variables

Local variables can be of any of the data types listed in "Assigning Data Types" on page 12. Two of these types require special consideration:

instance handle      The identification of an instance of a class. The implementation of this type is entirely dependent on the architecture; hence, its form is unknown to the user of the OAL.

instance handle set    A set of instance handles.

Because instance handles and instance handle sets are obtained by selection from existing instances, the following situations can arise: (1) a local variable of type instance handle may not contain a valid reference to an instance, and (2) a local variable of type instance handle set may refer to an empty set. These situations can be detected by using the supplied unary set operators.

### 4.3.1 Notes

Attributes cannot be of type instance handle or instance handle set.

Strictly speaking, a local variable is not of type instance handle, but rather of type instance handle for a particular class. Any attempt to use an instance handle for one class in the context of another is an error.

In any action within an instance statechart, a special instance handle called `self` is always available. `Self` is always defined as a handle to the instance of the class that is executing the current action. The value of `self` cannot be changed by an action.

The instance handle `self` has no meaning inside of an action for a class-based operation, a bridge operation, or a function. `Self` is not defined for a class statechart.

In any action within an instance-based operation or mathematically-dependent attribute, `self` is always available. `Self` is always defined as an instance handle to the instance of a class against which the operation is being executed, or the instance for which the attribute is being read. The value of `self` cannot be changed by an action.

## 4.4 Assigning Data Types

The data type names used in this manual have been chosen for readability. The table lists the correspondences between the data type names used here and in BridgePoint Model Builder.

in this manual	in Model Builder
integer	integer
real	real
boolean	boolean
string	string
date	date
timestamp	timestamp
unique ID	unique_ID
state	state<State_Model>
timer handle	inst_ref<Timer>
instance handle	inst_ref<Object>
instance handle set	inst_ref_set<Object>
event instance	inst<event>

**Table 4-2:** Names of data types used in this manual and their corresponding names used in the BridgePoint Model Builder.

## 4.5 Variable Initialization

Some situations arise where a variable may possibly be declared without being assigned a value. An obvious situation where this occurs is when an instance of a class with attributes is created. Though the attributes should not be accessed before they are assigned, there is nothing preventing the user from attempting to do so.

For the purposes of the Model Verifier, unassigned variables are considered UNDEFINED. The user should not intentionally read the data from UNDEFINED variables.

For the purposes of the translated code, the value of unassigned variables lies entirely in the realm of the software architecture defined by the model compiler. Even so, use of an unassigned variable as a read value or in an expression should be avoided.

### 4.5.1 Examples

```
// Reading an uninitialized attribute
create object instance d of DOG; // Attributes of d are not initialized
dog_name = d.name; // Error! Cannot read uninitialized value

// Using an uninitialized instance reference
select many f_set from instances of fish; // Select an empty instance set
for each f in f_set
    // statement block
end for;
// Instance reference f is available in this scope
// and may be uninitialized if f_set was empty.
fish_type = f.type; // Warning! f may be uninitialized.
```

## 4.6 Scoping

The scope of a variable is defined as the block of code in which the variable may be accessed. A block of code can be the entire OAL for the given action, or it may be a <statements> block within a control logic structure.

Each control logic structure contains at least one new scope. All variables that were accessible in the scope containing the structure are also accessible in the block or blocks contained by the structure, essentially causing the contained scopes to inherit variables from the parent scope. Any variables declared within a given control logic block fall out of scope when execution exits the block.

Control logic structures may contain multiple scopes, either by repeated nesting of new structures or by using the `elif` or `else` constructs in an `if` structure. When nesting of control logic is used, each new structure defines a new scope. In an `if` statement, each `elif` or `else` structure contained within the `if` block defines a new scope, and each new scope inherits the scope of the block containing the `if` statement.

### 4.6.1 Notes

A local variable is implicitly declared at the moment it is assigned, with a scope limited to the current block.

Local variables have a maximum scope of the entire OAL for the current action.

In the `for` statement, the local variable declared by <instance handle> has the same scope as the block containing the `for` statement.

The `where` clause has a special variable, `selected`, that has scope limited to the <where expression>.

The scope of `self` for an instance's action, operation, or attribute, is the entire OAL for the current action.

## 4.6.2 Example

```
// begin action
// scope1 - global scope for this action.  Variables declared here
// are accessible anywhere in this action.
delta = self.destination - self.current_position;
if (delta == 0)
    // scope2 - Variables declared here are only accessible
    // within if statement.
    spin_spot = CARPIO::carousel_spin(car_id:self.carousel_ID);
end if; // All variables declared in scope2 are not accessible
        // after the end if.
select many rows from instances of ROW;
for each row in rows
    // scope3 - Variables declared here are only accessible within
    // for each statement.
    st = row.sampling_time;
end for; // All variables declared in scope3 are not accessible after
        // the end for.
// row is available in the scope containing the for each statement
// (scope1).

if (delta <= 2)
    // scope4 - Variables declared here are accessible within this
    // if block and in scope4.1 and scope4.2.
    if (CARPIO::angle(car_id:self.carousel_ID) == 30)
        // scope4.1 - Variables declared here are only accessible
        // within this if block.
        end if;
    if (CARPIO::angle(car_id:self.carousel_ID) == 60)
        // scope4.2 - Variables declared here are only accessible within
        // this if block.
        end if;
    end if;
end if;
// end action
```



# 5

## CONTROL STRUCTURES

---

### 5.1 *If Construct*

#### 5.1.1 *Syntax*

```
// Note that there is no semi-colon following the
// "if <boolean expression>"
if <boolean expression>
    <statements> // Executed if <boolean expression> is TRUE
end if;

if (<boolean expression>)
    <statements> // Executed if above boolean expression evaluates to TRUE
elif (<boolean expression>)
    <statements> // Executed if above boolean expression evaluates to TRUE
                // and previous boolean expression is FALSE
else
    <statements> // Executed if both boolean expressions evaluate to FALSE
end if;
```

<boolean expression> is an expression evaluating to TRUE or FALSE.

#### 5.1.2 *Notes*

The if construct may contain as many elif clauses as desired.

Only one else clause may be used, and it must appear at the end of the if construct.

### 5.1.3 Example

The following example shows an if/elif/else construct:

```
// Assign x with a different number for each name.
if (name == "John")
    x = 1;
elif (name == "Bill")
    x = 2;
elif (name == "Michael")
    x = 3;
else
    // If not a known name, assign x to 4.
    x = 4;
end if;
```

This example shows nested if constructs:

```
// Carousel: Going
self.destination = rcvd_evt.destination;
delta = self.destination - self.current_position;
if ( delta == 0 )
    generate C2:there to self;
else
    select any probe from instances of SP
    where (selected.current_position == "down");
    if (not_empty probe)
        generate C2:there to self;
    else
        spin_spot = CARPIO::carousel_spin(
            car_id:self.carousel_ID, destination:delta );
    end if;
end if;
```



## 5.2 For Each Loop

The for each loop allows for the iteration over a set of instance handles in an instance handle set.

### 5.2.1 Syntax

```
for each <instance handle> in <instance handle set> // Note no semi-colon
    <statements>
end for;
```

<instance handle> is a local variable referring to a single instance.

<instance handle set> is a local variable referring to a set of instance handles.

### 5.2.2 Notes

The statements in the for each construct are executed once against each instance in <instance handle set>.

The order in which the particular instances are processed is undefined.

☹ Because the statements in the for each construct can, in principle, be executed in parallel (as when instances are dispersed over multiple processors), the concept of a loop counter is undefined. Consequently, the analyst should not attempt to defeat this restriction.

### 5.2.3 Example

```
// C is the keyletter for the Child object.
// children is an implicitly typed variable of <instance handle set> of C.
select many children from instances of C;
for each child in children
    generate C1:'time for bed' () to child;
end for;
```

## 5.3 While Loop

The while construct is used to sequentially execute the code it contains for as long as the condition is evaluated as TRUE.

### 5.3.1 Syntax

```
while (<boolean expression>) // Note no semi-colon
    <statements>
end while;
```

<boolean expression> is an expression evaluating to TRUE or FALSE

<statements> are zero or more OAL statements.

### 5.3.2 Note

When the while loop is executed, the statements in the while construct are executed consecutively as long as the <boolean expression> evaluates to TRUE.

### 5.3.3 Example

```
// Create 20 doors with IDs 1-20
i = 1;
while (i <= 20)
    create object instance d of DOOR;
    d.ID = i;
    i = i + 1;
end while;
```

## 5.4 Break

The break statement allows the early termination of both for each and while loops. This can have some significant performance implications in the case of large loops that need not step through the entire iteration.

### 5.4.1 Note

The break statement only applies to the current for each or while loop containing it. To break out of nested loops, the break must be repeated for each loop construct the user wishes to exit.

### 5.4.2 Example

```
// Create and relate a B to every A while CTL says to keep
// creating.
while (CTL::create())
    breakout = FALSE;
    for each a in aset
        // If this a has name equal to "Jeff", break out of
        // for each loop.
        if (a.name == "Jeff")
            breakout = TRUE;
            break;
        end if;
        // Create and relate a new b to the given a.
        create object instance b of B;
        relate b to a across R1;
    end for;
    // If "Jeff" was found, break out of while loop also.
    if (breakout)
        break;
    end if;
end while;
```

## 5.5 Continue

The continue statement causes the next iteration of the enclosing for each or while loop to begin, avoiding execution of the loop's remaining code.

### 5.5.1 Note

The continue statement only applies to the current loop containing it.

### 5.5.2 Example

```
// Create and relate a B to each A, except for As with ID of 13.
for each a in aset
    // If a.ID is 13, don't create and relate a B to it and
    // continue to the next.
    if (a.ID == 13)
        continue;
    end if;
    create object instance b of B;
    relate b to a across R1;
end for;
```

## 5.6 Nested Control Logic

Control logic may be nested to any depth.

### 5.6.1 Example

```
// Send a 'time for bed' event to all children 5 and under.
select many children from instances of C;
for each child in children
  if (child.age <= 5)
    while (child.awake)
      generate C1:'time for bed' () to child;
      if (not lights.out)
        generate C2:'turn off lights' () to child;
      end if;
    end while;
  end if;
end for;
```



## CLASS MANIPULATIONS

---

### 6.1 Creating Instances

Creation of an instance of a class is achieved by use of the create statement.

#### 6.1.1 Syntax

```
create object instance <instance handle> of <keyletter>;  
create object instance of <keyletter>;
```

<keyletter> is the keyletter of a class in the model.

#### 6.1.2 Notes

The <instance handle> returned is the handle of the newly created instance of class <keyletter>. The handle can be used only to refer to an instance of class <keyletter>.

The <instance handle> in the create statement cannot be self.

The value of all identifying attributes must be set by the analyst before completion of the action in which an instance is created. Note that identifying attributes of type unique ID cannot be assigned values, as they are initialized by the system.

☺ All unconditional relationships that involve the newly created instance should be satisfied before completing the action in which the instance is created. This can be done either by directly relating the associated instance or by generating an event that will cause the newly created instance to be properly related.

### 6.1.3 Examples

```
// Create instances of autosampler classes
create object instance car of C;
create object instance row of ROW;

create object instance of SP; // no instance handle needed
```

## 6.2 Selecting Instances

The select statement can be used to assign an instance or set of instances to either an instance handle or a instance handle set respectively. An optional where clause can be used at the end of the select statement to limit the selection. Within the where clause, the selected instance handle refers to each of the instances in the entire set defined by <keyletter>. The instance handle selected is meant to be used as an instance handle in a boolean comparison to form the where expression. The instance or set of instances returned match the criteria of the where expression, and may be empty.

### 6.2.1 Syntax

```
select any <instance handle> from instances of <keyletter>;
select many <instance handle set> from instances of <keyletter>;
select any <instance handle> from instances of <keyletter> where <where
expression>;
select many <instance handle set> from instances of <keyletter> where
<where expression>;
```

<instance handle> is the handle for an instance of the class specified by <keyletter>.

<instance handle set> is a set of handles for all selected instances of the class specified by <keyletter>.

<where expression> is a type of boolean expression using selected keyword.



## 6.2.2 Notes

If the optional where clause is used, the returned instance or set of instances meet the criteria of the where expression. This implies that the instance handle may be empty, or the instance handle set may be empty if no instance fulfills the criteria.

If the select any form is used, an arbitrary instance will be obtained from the selected set.

If the select many form is used then the entire set of instances will be obtained.

If the select any ... where form is used, an arbitrary instance that fulfills the <where expression> will be obtained.

If the select many ... where form is used then the set of instances that fulfill the <where expression> will be obtained.

The instance handle selected is valid only within <where expression>.

## 6.2.3 Example

```
// Select an arbitrary instance.
select any dp_one from instances of DP;

// Select all instances.
select many dp_set from instances of DP;

// Select an instance of DP whose available attribute is TRUE.
select any dp_avail from instances of DP where selected.available == TRUE;

// Select a set of instances from DP whose available attribute is FALSE.
select many dp_unavail_set from instances of DP
  where selected.available == FALSE;
```

Making selections across relationships is describe in "Relationship Navigation" on page 40.

## 6.3 Writing Attributes

Attributes of instances may be set to specified values by use of the assign statement.

### 6.3.1 Syntax

```
[assign] <instance handle>.<attribute> = <expression>;
```

<instance handle> is a handle to an instance of a class.

<attribute> is the name of an attribute of the class.

<expression> is either a boolean, string, or arithmetic expression.

The assign statement takes the value in <expression> and assigns it to the attribute <attribute> for the instance specified by <instance handle>.

The assign keyword is optional.

### 6.3.2 Notes

The <expression> must evaluate to the data type of <attribute>, unless <attribute> is either a real or an integer, in which case <expression> can be either a real or integer value.

A value cannot be assigned to a referential attribute. Relationships must be maintained via the relate and unrelate constructs.

A value cannot be assigned to an attribute of type unique ID, as such an attribute is initialized by the system when the instance is created.

Mathematically-dependent attributes cannot be written to, only read. The value of a mathematically-dependent attribute must be set within the action of the attribute. See "Writing Mathematically-Dependent Attributes" on page 29.

### 6.3.3 Example I

```
// Account is a class with attributes branch, account_number,  
// and balance. Assume new_account, this_branch, and initial_deposit  
// are event supplemental data items.  
  
// First, create a new instance.  
create object instance my_account of ACCT;  
  
// Now set attribute values using the returned instance handle.  
my_account.branch = rcvd_evt.this_branch;  
my_account.account_number = rcvd_evt.new_account;  
my_account.balance = rcvd_evt.initial_deposit;
```

### 6.3.4 Example II

```
// Create and initialize a row in the autosampler carousel.  
create object instance row of ROW;  
relate row to car across R1;  
row.radius = 10;  
row.current_sampling_position = 0;  
row.maximum_sampling_positions = 5;  
row.sampling_time = 5000;  
row.needs_probe = false;
```

## 6.4 Writing Mathematically-Dependent Attributes

Mathematically-dependent attributes are attributes that have their value derived from other modeled elements. It is not possible to directly write to a mathematically-dependent attribute as described above. Instead, OAL must be used to specify the derived value. When the attribute is read in an action, the value of the attribute is calculated from the OAL specified in the action for the mathematically-dependent attribute.

## 6.4.1 Syntax

From within the action of a mathematically-dependent attribute:

```
[assign] self.<attribute> = <expression>;
```

<attribute> is the name of the mathematically-dependent attribute.

<expression> is either a boolean, string, or arithmetic expression.

The assign statement takes the value in <expression> and assigns it to the attribute <attribute> for the instance specified by self.

The assign keyword is optional.

## 6.4.2 Notes

The <expression> must evaluate to the data type of <attribute>, unless <attribute> is either a real or an integer, in which case <expression> can be either a real or integer value.

A value cannot be assigned to a referential attribute. Relationships must be maintained via the relate and unrelate constructs.

A value cannot be assigned to an attribute of type unique ID, as such an attribute is initialized by the system when the instance is created.

The action parsing routine for attributes checks to see if the variable self is written somewhere in the state action, and if not, a parse error is reported.

Care should be taken to make sure that all paths inside an action actually set the attribute.

The param and return keywords are not supported in the action of the attribute.

### 6.4.3 Example

```
// Mathematically-dependent attribute (MDA) action
self.volume = self.length*self.width*self.height;

// Action reading the MDA volume
v = cube.volume;
```

## 6.5 Reading Attributes

A class attribute may be referenced in an expression using the form:

### 6.5.1 Syntax

`<instance handle>.<attribute>`

`<instance handle>` is a handle to an instance of a class.

`<attribute>` is the name of an attribute of the class.

### 6.5.2 Notes

You may read the value of any attribute, including referential attributes.

`<instance handle>.<attribute>` is an expression and can be used in any OAL construct specifying an expression.

☺ When you wish to obtain an associated instance, use the relationship navigation constructs rather than reading (a succession of) referential attributes. This ensures that code generators can detect the purpose of the read and therefore produce accurate and effective translation.

### 6.5.3 Example

```
// Create new instance and get handle.
Create object instance myrobot of R;
// Use the instance handle to read attribute values.
myx = myrobot.x_position;
myy = myrobot.y_position;

// Position the row for sampling.
select one car related by self->C[R1];
self.next_sampling_position = self.current_sampling_position + 1;
next =
  ROW::convert_dest( radius:self.radius,
    next_sampling_position:self.next_sampling_position );
generate C1:go(destination:next) to car;
```

## 6.6 Deleting Instances

### 6.6.1 Syntax

```
delete object instance <instance handle>;
```

### 6.6.2 Notes

This statement deletes the instance specified by <instance handle>.

When an instance of a class is deleted, it is no longer available to the domain where the class is defined. However, sophisticated software architectures can be imagined which support the notion that the instance is kept for logging purposes although the defining domain cannot see it.

☺ Depending on the architecture, deleting an instance may or may not be sufficient to specify deletion of any attached relationships. Because certain architectures may fail if such dangling relationships are used at run time, we recommend that the analyst explicitly delete relationships before deleting the participating class instances.

### 6.6.3 Example

```
// Delete every instance of DG with name equal to Fido.
select many dogs from instances of DG where (selected.name == "Fido");
for each dog in dogs
    select one owner related by dog->OWN[R23];
    unrelate dog from owner;
    delete object instance dog;
end for;
```





---

## **7.1 Relationship Specifications**

A relationship specification identifies exactly which relationship is required to be created, navigated, or deleted.

### **7.1.1 Syntax**

```
R<number>  
r<number>  
R<number>.<relationship phrase>  
r<number>.<relationship phrase>
```

<number> the number of the relationship as shown on the class diagram.(e.g., R1).

<relationship phrase> is the text that appears at the destination end of the relationship, enclosed in tick marks and contained on a single line.

### **7.1.2 Note**

Either R or r may be used when referring to a relationship.

### 7.1.3 Examples

```
r5
R10.'owns'
r10.'is owned by'
R22.'uses'
R1.'Is rotated by'
R1.'Contains'
R2.'Is assigned to'
```

## 7.2 Creating an Instance of a Relationship

### 7.2.1 Syntax

```
relate <source instance handle> to <destination instance handle> across
<relationship specification>;
```

```
relate <source instance handle> to <destination instance handle> across
<relationship specification> using <associative instance handle>;
```

<source instance handle> is the handle of the first class instance to be related.

<destination instance handle> is the handle of the second class instance to be related.

<relationship specification> is the specification of the relationship from the source class to the destination class. This can be any of the forms described in the previous section.

<associative instance handle> is the handle of an existing class instance that is used as the associative class instance for this relationship instance.

## 7.2.2 Notes

The relationship specification should be framed as if navigating from source class to destination class.

The source, destination, and associative instance handles may be `self`.

If an attempt is made to relate two instances via the same relationship more than once, this is regarded as a run time error by the BridgePoint Model Verifier unless the relationship is (M:M)-M.

The using `<associative instance>` form is used when an associative relationship is being instantiated. The associative instance must have already been created before the relationship is instantiated.

## 7.2.3 Examples

```
select any dp_inst from instances of DP;
select any d_inst from instances of D;
relate dp_inst to d_inst across R1;

select any a_inst from instances of A;
select any b_inst from instances of B;
create object instance c_inst of C;
relate a_inst to b_inst across R1 using c_inst;

// State 3. "Assigning Probe to Row"
select any row from instances of ROW
  where ( selected.needs_probe == true );
select any probe from instances of SP
  where ( selected.available == true );
probe.available = false;
row.needs_probe = false;
create object instance assignment of PA;
relate row to probe across R2 using assignment;
generate PA_A3:probe_assigned() to PA class;
generate ROW2:probe_assigned() to row;
```

## 7.3 Deleting an Instance of a Relationship

### 7.3.1 Syntax

```
unrelate <source instance handle> from <destination instance handle> across  
<relationship specification>;
```

```
unrelate <source instance handle> from <destination instance handle> across  
<relationship specification> using <associative instance handle>;
```

<source instance handle> is the handle of the first class instance to be unrelated.

<destination instance handle> is the handle of the second class instance to be unrelated.

<relationship specification> is the specification of the relationship from the source to the destination class.

<associative instance handle> is the handle of the associative class instance that captures the relationship instance.

### 7.3.2 Notes

The relationship specification should be framed as if navigating from the source class to the destination class.

An attempt to unrelate two instances that are not related by the specified relationship is regarded as a run time error by the BridgePoint Model Verifier.

The source, destination, and associative instance handles may be `self`.

☺ If an associative relationship is unrelated then the associative class instance(s) will not be deleted. The analyst must specify this explicitly.

☺ If an unconditional relationship is deleted, instances of participating classes will not automatically be deleted to remain consistent with the Class Diagram. It is the responsibility of the analyst to ensure that the Class Diagram is respected. This applies equally to super/subtype relationships.

### **7.3.3 Examples**

```
unrelate a_inst from b_inst across R1;  
unrelate a_inst from b_inst across R1 using c_inst;  
delete object instance c_inst;
```

## 7.4 Relationship Navigation

Relationship navigation is the function whereby relationships specified on the Class Diagram are read in order to determine the instance or set of instances that are related to an instance of interest.

### 7.4.1 Syntax

```
select one <instance handle> related by <start> -> <relationship link> ->
... <relationship link>;
```

```
select any <instance handle> related by <start> -> <relationship link> ->
... <relationship link>;
```

```
select many <instance handle set> related by <start> -> <relationship link>
-> ... <relationship link>;
```

```
select one <instance handle> related by <start> -> <relationship link> ->
... <relationship link> where <where expression>;
```

```
select any <instance handle> related by <start> -> <relationship link> ->
... <relationship link> where <where expression>;
```

```
select many <instance handle set> related by <start> -> <relationship link>
-> ... <relationship link> where <where expression>;
```

<start> is an <instance handle set> or <instance handle> obtained from a previous select statement.

<relationship link> is a <keyletter>[<relationship specification>], where the square brackets are literal and do not indicate optional text.

<keyletter> is the keyletter of the class reached by the specified relationship.

<relationship specification> is the specification of the relationship from the source to the destination class.

<where expression> is a type of boolean expression using the selected keyword.

## 7.4.2 Notes

A *relationship link chain* is the sequence of `<relationship link>`'s used to specify the path from the starting instance or set of instances to the destination.

Use the `select one` form if at most one instance handle can be returned by navigating the relationship link chain.

Use the `select any` or `select many` form if more than one instance handle can be returned by navigating the relationship link chain. `Select any` returns a single instance, and `select many` returns all instances that meet the selection criteria.

The `select any` form returns the instance handle of an arbitrary instance of the class at the end of the relationship link chain.

The `select many` form returns an instance handle set containing all the instances of the class at the end of the relationship link chain.

The `select any ... where` form returns the instance handle of an arbitrary instance of the class at the end of the relationship link chain that fulfills the `<where expression>` criteria.

The `select many ... where` form returns an instance handle set containing all the instances of the class at the end of the instance chain that fulfill the `<where expression>` criteria.

The relationship phrases in the relationship link chain must be given in the direction of navigation.

If the starting `<instance handle>` or `<instance handle set>` is empty, then the result will be considered a run time error.

The returned `<instance handle>` or `<instance handle set>` can be empty if any of the relationships in the chain are conditional in the direction of navigation.

If the optional `where` clause is added, the returned instance or set of instances will meet the criteria of `<where expression>`. This implies that the instance handle or the instance handle set may be empty if no instance(s) matched.

### **7.4.3 Example I**

```
select one cat related by owner->C[R1];
select any dog related by owner->D[R2];
select many dogs related by owner->D[R2];

select any assignment from instances of PA here ( selected.probe_ID ==
  self.probe_ID );
select any dog related by owner->D[R2] where ( selected.name == "Fido" );
select many dogs related by owner->D[R2] where selected.color == "black";
```

### **7.4.4 Example II**

```
select any student from instances of STU;
select many major_courses_offered related by
  student->PROF[R34]->DEPT[R23]->COUR[R40];
```



---

## 8.1 Receiving Event Data

The keyword `rcvd_evt` is the name of a structure containing all of the supplemental data items received with an event.

### 8.1.1 Syntax

```
rcvd_evt.<supplemental data item>
```

`<supplemental data item>` is the name of the data item.

### 8.1.2 Note

`rcvd_evt.<supplemental data item>` is an `<expression>` and so can be used in any OAL construct specifying an expression.

### 8.1.3 Example

```
select any robot from instances of R;  
robot.from = rcvd_evt.source;  
robot.to = rcvd_evt.destination;  
  
//  
self.destination = rcvd_evt.destination;
```

## 8.2 Event Generation

### 8.2.1 Syntax

```
generate <event label> to <target>;  
generate <event label>:<event meaning> to <target>;  
generate <event label> (<event parameters>) to <target>;  
generate <event label>:<event meaning> (<event parameters>) to <target>;  
generate <instance handle>.<attribute>;
```

<event label> is <keyletter><event number>.

<event meaning> is the meaning of the event, enclosed by tick marks as in 'turn off the light'; the tick marks may be omitted if the event meaning contains no spaces.

<event parameters> provides the supplemental data items (if any) to be carried by the event. Each data item is given in the form <supplemental data item>:<expression>. When multiple supplemental data items are required, separate the <supplemental data item>:<expression> pairs by commas. If there are no supplemental data items, the parentheses may be omitted.

<supplemental data item> is the name of a data item to be sent with the event.

<expression> is a string, arithmetic, boolean, simple, or compound expression. The data type of the expression must match the data type defined for the given data item.

<target> is specified in a variety of ways, depending on the destination of the event.

For an event directed to an existing instance of a class, <target> is an instance handle.

For a creation event, <target> is: <keyletter> creator.

For an event directed at a single-instance assigner, <target> is: <keyletter> class.

For an event directed at an external entity, <target> is the external entity's keyletters.

<instance\_handle> is a handle to an instance of a class.

<attribute> is the name of an attribute of the class.

## 8.2.2 Notes

The `to` clause provides all the information necessary to identify the destination of the event.

If an event has no supplemental data items, the empty parentheses around `<event parameters>` may be omitted.

Supplemental data items may appear in any order in `<event parameters>`.

All supplemental data items defined for the event must be supplied.

The `<event meaning>` field is optional. It must be enclosed in tick marks if it contains spaces, and it must be contained on a single line.

If `<event meaning>` is not used, the colon after `<event label>` must be omitted.

## 8.2.3 Examples

```
// event to existing instance
// State 1. "Up"
self.current_position = "up";
select one row related by self->ROW[R2];
generate ROW4:sample_complete() to row;

// creation event
generate S1:'Create sale' (dept:dept_no, amount:sale_value) to S creator;

// event to assigner
generate PA_A1:row_needs_probe() to PA class;

// event to external entity
generate PIO7:'Motor start' (motor_no:motor_id) to PIO;
```

## 8.3 Event Pre-creation

An event may be created without sending it by using the `create event` statement. This statement should be used only

☺ to create an event for an analysis timer.

☺ as directed by the rules of the particular architecture onto which you plan to translate your models.

### 8.3.1 Syntax

```
create event instance <event instance> of <event label> to <target>;
```

```
create event instance <event instance> of <event label>:<event meaning> to  
<target>;
```

```
create event instance <event instance> of <event label> (<event  
parameters>) to <target>;
```

```
create event instance <event instance> of <event label>:<event meaning>  
(<event parameters>)to <target>;
```

<event label>, <event meaning>, <event parameters> and <target> are as defined in "Event Generation" on page 44.

<event instance> is a local variable of type event instance.

### 8.3.2 Notes

The local variable <event instance> can be used to refer to an event instance of any type.

Event instances can be assigned to attributes of classes that are of type event instance.

Please refer to the notes in "Event Generation" on page 44.

## 8.4 Sending a Pre-created Event

Event instances may be sent using the `generate` statement.

### 8.4.1 Syntax

```
generate <event instance>;
```

```
generate <instance handle>.<attribute>;
```

`<event instance>` is a local variable of type `event instance`.

`<instance handle>` is a handle to an instance of a class.

`<attribute>` is an attribute of the class.

### 8.4.2 Notes

Sends a pre-created event to its intended recipient.

☺ This feature is provided for use with analysis timers. These timers will eventually be replaced with delayed events, at which time, support for pre-created events will likely be removed.



---

## 9.1 Simple Expressions

Simple expressions are single unary or binary operations. An expression is not a complete OAL statement, but is evaluated as part of a full OAL statement such as `assign`, `if`, `where`, etc. Logical binary operators `and` and `or` are supported for both compound and simple expressions.

### 9.1.1 Syntax

```
<read value>  
<unary operator> <read value>  
<read value> <binary operator> <read value>
```

`<read value>` is a constant, a local variable, the attribute of a class, a supplemental data item received from an event, an operation invocation, a bridge invocation, or a function invocation. It can also be a parameter specified with the `param` keyword in the action of a bridge operation, function, or operation.

`<unary operator>` is any unary operator appropriate for the data type to which the expression evaluates. For boolean read values, the unary operator is `not`. For arithmetic read values, the unary operators are `+` and `-`. For instance handle and instance handle set read values, the unary operators are `empty`, `not_empty`, and `cardinality`.

`<binary operator>` is any binary operator appropriate for the data types to which the expressions evaluate. For boolean read values, the binary operators are `and` and `or`. For arithmetic read values, the binary operators are `+`, `-`, `*`, `/`, and `%`. For instance handle and instance handle set read values, the binary operators are `==` and `!=`. For string read values, the binary operator is `+`.

## 9.1.2 Example

```
not (CHK::get_status())
x + y
name == "Jeff"
"Bridge" + "Point"
cust1.age - cust2.age
```

## 9.2 Compound Expressions

Compound expressions can be used to combine simple expressions, allowing for multiple tests and more complex assignment arithmetic. Logical binary operators and or are supported for both compound and simple expressions.

### 9.2.1 Syntax

```
<operator> <expression>
<read value> <operator> <expression>
<expression> <operator> <read value>
<expression> <operator> <expression>
```

<expression> is a simple or a compound expression.

<operator> is any operator appropriate for the data types to which the expressions evaluate.

<read value> is a constant, a local variable, the attribute of a class, a supplemental data item received from an event, an operation invocation, a bridge invocation, or a function invocation.



## 9.2.2 Notes

The analyst can depend on the following rules regarding the order of evaluation of expressions:

- Parentheses can be used to override all other ordering rules.
- Standard mathematical precedence governs the order of evaluation for all mathematical operations.
- All subexpressions with operators of equal precedence are evaluated from left to right, starting with the operators of highest precedence. This is repeated until the compound expression has been completely evaluated.
- A short-circuit with regard to compound expressions means that an expression can be fully evaluated based upon the value of one of its subexpressions. Whether or not short-circuiting occurs depends entirely on the implementation of the software architecture or the simulator being used. The analyst should therefore avoid writing OAL that depends on short-circuiting of expressions.

## 9.2.3 Examples

```
// examples of compound expressions:  
not (arm.available and servo.on)  
2 * (x + y) + TIM::timer_remaining_time(timer_inst_ref:timer_1)  
(a + b) / (c - d)
```

```
// examples of OAL statements using  
// compound expressions:  
if ((i == 1) AND (name == "Doug"))  
    assign x = 0.5 * (y + z);  
end if;  
  
x = x * ((x + 1) / (x + 2));
```

## 9.3 Arithmetic Expressions

Arithmetic expressions are defined for real and integer data types only. These data types may be mixed for any given expression. Multiplicative operators are \*, /, and %. Additive operators are + and -. Multiplicative operators take precedence over additive operators. Parentheses may be used to force precedence in arithmetic expressions.

### 9.3.1 Syntax

```
<unary arithmetic operator> <expression>  
<expression> <binary arithmetic operator> <expression>
```

<expression> is any of the following that evaluates to a real or integer value: numeric constant, local variable, attribute of a class, simple expression, compound expression, operation invocation, bridge invocation, function invocation, or supplemental data item received from an event.

<unary arithmetic operator> is + or -.

<binary arithmetic operator> is +, -, \*, /, or % (remainder from arithmetic division).

### 9.3.2 Note

If any data item in the expression is real, the expression will evaluate to a data type of real.

### 9.3.3 Examples

```
-27  
2 + 2  
(x + y) / 2  
0.707 * voltage  
(plane.offset + ALT::get_altitude())
```

## 9.4 Boolean Expressions

A boolean expression is any expression that evaluates to either a TRUE or FALSE value. Boolean expressions are often used for comparison in statements like `if` and `while`, and also in `where` clauses. Although boolean expressions usually contain other expression types (such as arithmetic or string expressions), they can also be used to compare time values, handles, and unique IDs. There is also one unary operator, `not`, which can be used to logically negate a boolean expression.

### 9.4.1 Syntax

```
not <boolean expression>  
<expression> <boolean operator> <expression>  
<time value> <boolean operator> <time value>  
<handle> <boolean operator> <handle>  
<unique_id> <boolean operator> <unique_id>
```

`<expression>` is any expression, simple or compound. Both expressions must evaluate to the same type, either boolean, arithmetic, or string.

`<boolean expression>` is a simple or compound expression that evaluates to a boolean value.

`<boolean operator>` is a logical operator (refer to "Table 9-1" on page 54).

`<time value>` a date or a timestamp variable (or an operation, bridge, or function invocation that returns a date or a timestamp).

`<handle>` an instance handle, instance handle set, or a timer handle (or an operation, bridge, or function invocation that returns a handle to a timer).

`<unique id>` a variable of type unique ID (or an operation, bridge, or function invocation that returns a unique ID).

## 9.4.2 Note

The left and right values of a binary boolean expression must evaluate to the same data type, with the exception of integer and real.

## 9.4.3 Examples

```
x == 1
id != "abc"
CTL::error() or flag
(account.balance == 0.00) and ((TIM::get_current_time() - last_pay_time) >=
    max_wait)
```

Logical Operator	Meaning	Valid Data Types
==	equals	integer, real, boolean, date, timestamp, string, instance handle, unique ID, handle set, timer handle
!=	does not equal	integer, real, boolean, date, timestamp, string, instance handle, unique ID, handle set, timer handle
<	less than	integer, real, date, timestamp, string
>	greater than	integer, real, date, timestamp, string
<=	less than or equal to	integer, real, date, timestamp, string
>=	greater than or equal to	integer, real, date, timestamp, string
and	logical and	boolean
or	inclusive logical or	boolean
not	logical negation	boolean

**Table 9-1:** The logical operators defined in the Object Action Language. Note that certain operators are valid for certain data types only

## 9.5 String Expressions

A string expression is any expression that evaluates to a string value. String expressions can be either a simple string or a concatenation of one or more simple strings.

### 9.5.1 Syntax

```
<simple string>  
<simple string> + ... + <simple string>;
```

<simple string> is any of the following that evaluates to a string value: string constant, local variable, attribute of a class, operation invocation, bridge invocation, function invocation, or a supplemental data item received from an event.

### 9.5.2 Examples

```
"Hello, world!"  
"Executable" + "-" + "UML"  
cust.first_name + " " + cust.last_name  
CHS::get_date_string(date:TIM::current_date())
```

## 9.6 Where Expressions

A where expression is a special type of boolean expression used in a select statement. The instance handle selected is valid only within the where expression. The selected keyword should be used as an instance reference to access the instances of the given set for the select statement containing the where expression. The where expression must evaluate to a boolean value, and must use the selected keyword.

### 9.6.1 Note

The where expression can only be used in the where clause of a select statement.

## 9.6.2 Examples

```
select any firstname in EMP where selected.name == "Bob";
select many accounts in ACC where (selected.status == "Ok") and
(selected.balance > (min_bal + 200));
```

```
// Use where clause to find a particular probe.
select any probe from instances of SP
  where selected.probe_ID == param.probe_id;
generate SP3:probe_in_position to probe;
```

## 9.7 Assignment of Variables

Calculations are performed using the following form of the assign statement:

### 9.7.1 Syntax

```
[assign] <boolean var> = <boolean expression>;
[assign] <arithmetic var> = <arithmetic expression>;
[assign] <string var> = <string expression>;
```

<boolean expression> is an expression evaluating to TRUE or FALSE.

<arithmetic expression> is an expression evaluating to a real or an integer value.

<string expression> is an expression evaluating to a string value.

<boolean var> is a boolean variable or a boolean attribute of a class instance.

<arithmetic var> is a real or integer variable or a real or integer attribute of a class instance.

<string var> is a string variable or a string attribute of a class instance.

## 9.7.2 Notes

Arithmetic expressions are defined for real and integer data types only.

If `<arithmetic var>` is a local variable that is being assigned for the first time, it will be of type integer if `<arithmetic expression>` evaluates to type integer, and of type real if `<arithmetic expression>` evaluates to type real.

Once an `<arithmetic var>` has been assigned, it will not change data type from real to integer or from integer to real. Real and integer variables can, however be assigned integer or real values respectively.

If a real value is assigned to an integer variable, the fractional component is truncated.

The actual precision and truncation rules for arithmetic calculation depend on the software architecture and implementation domains in use.

The `assign` keyword is optional.

## 9.7.3 Examples

```
assign x = 1;  
pass = TRUE;  
assign name = "Rover";  
f = RTR::get_frequency() + 100;
```

## 9.8 Constants

In many of the examples, constants have been used as parts of expressions. While this serves well for the purposes of illustration, it should be noted that most analysis models require minimal use of constants since such data is more commonly stored as attributes of specification classes.

## 9.8.1 Syntax

The syntax depends on the base data type:

Integer:	1, 42, -127, etc.
Real:	1.0, 4.5, -56.0, etc.
String:	"string"
Boolean:	TRUE, FALSE

**Table 9-2:** Syntax of Constant Data

## 9.8.2 Notes

Constants may be defined for the above data types only.

A constant may be used in any construct requiring an expression.

## 9.9 Additional Unary Operators

Three set operators have been provided to allow the analyst to determine the size of an instance handle set or whether or not an instance handle is defined. These operations may be performed anywhere an expression may be used.

### 9.9.1 Syntax

```
empty <handle>  
not_empty <handle>  
cardinality <handle>
```

<handle> is an <instance handle> or <instance handle set>.



## 9.9.2 Notes

`empty` and `not_empty` return a value of type `boolean`.

`cardinality` returns an integer value.

## 9.9.3 Example

```
select one d_inst related by self->D[R1];
if (not_empty d_inst)
    // Statements here protected against access to empty d_inst.
end if;
```



## OPERATIONS, BRIDGES, AND FUNCTIONS

---

### 10.1 Operation Invocation

The analyst may define class-based and instance-based operations as desired using the Operation Data Editor for a class under the Class Diagram. An operation invocation can be used as a stand-alone statement or it can be used in an expression.

#### 10.1.1 Syntax

As a operation expression:

```
<keyletter>::class-based operation name (<data item>:<expression>, ...)  
  
<instance handle>.<i>instance-based operation name (<data item>:  
<expression>, ...)
```

As a stand-alone operation assignment statement:

```
<variable> = <keyletter>::class-based operation name (<data item>:  
<expression>, ...);  
  
<variable> = <instance handle>.<i>instance-based operation name (<data  
item>:<expression>, ...);
```

<keyletter> is the keyletter of a class.

<instance handle> is a handle to an instance of a class.

<class-based operation> and <instance-based operation> are the name of the operation.

<data item> is the name of a data item defined as input for <class-based operation name> or <instance-based operation name>.

<expression> is a string, arithmetic, boolean, simple, or compound expression. The data type of the expression must match the data type defined for the given data item.

<variable> is a class attribute or a local variable.

## 10.1.2 Notes

An operation expression may be used anywhere an expression is valid (e.g., assignment statement read values, control logic expressions, etc.).

The stand-alone operation must always be a stand-alone statement and cannot be used as an expression within another statement.

Since an operation expression may be used anywhere an expression may be used, stand-alone operation assignment statements are not necessary. The user may simply use an operation expression as a read value and use an `assign` statement to assign the return value to <variable>.

Parentheses are required even if there are no data items.

If <variable> is a local variable that is being assigned for the first time, it will be of the same data type as the return value of <operation name>.

If the type of <variable> has already been established (that is, if it is the attribute of a class or a local variable that has been previously assigned), then either:

- <variable> must be of the same data type as the output value of the operation,
- or
- the output value of the operation is of type integer or real and <variable> is of type integer or real.

### 10.1.3 Example

```
// operation expressions without assigning the return value
DD::open(wait:20);
window.update(title:"Dialog");

// operation expression as an assignment statement read value
volume = DD::get_volume();

// operation expression within a while loop
while (MOD::status() != 1)
    value = this_mod.poll();
end while;

// stand-alone operation assignment statement
branch = TR::get_next_branch();
```

## 10.2 Bridge Invocation

### 10.2.1 Syntax

As a bridge expression:

```
<eekeyletter>::<bridge name> (data item:<expression>, ...)
```

As a stand-alone bridge assignment statement:

```
<variable> = <eekeyletter>::<bridge name> (<data item>:<expression>, ...);
```

<eekeyletter> are the keyletters of an external entity.

<bridge name> is the name of a bridge assigned to the external entity.

<data item> is the name of a data item input to <bridge name>.

<expression> is a string, arithmetic, boolean, simple or compound expression. The data type of the expression must match the data type defined for the given data item.

<variable> is a class attribute or a local variable.

## 10.2.2 Notes

It is strongly recommended that BridgePoint's external entities be used only to represent domains.

A bridge expression may be used anywhere an expression is valid (e.g., assignment statement read values, control logic expressions, etc.).

The stand-alone bridge assignment statement must always be a stand-alone statement and cannot be used as an expression within another statement.

Since a bridge expression may be used anywhere an expression may be used, stand-alone bridge assignment statements are not necessary. The user may simply use a bridge expression as a read value and use an assign statement to assign the return value to <variable>.

Parentheses are required even if there are no data items.

If <variable> is a local variable that is being assigned for the first time, it will be of the same data type as the output value of <bridge name>.

If the type of <variable> has already been established (that is, if it is the attribute of a class or a local variable that has previously been assigned), then either:

- <variable> must be of the same data type as the output value of the bridge,
- or
- the output value of the bridge is of type integer or real and <variable> is of type integer or real.

### 10.2.3 Example

```
// bridge expression without assigning the return value
OT::start();

// bridge expression as an assignment statement read value
cur_time = bridge TIM::current_time();

// bridge expression within the test part of an if statement
if (TIM::timer_add_time(timer_inst_ref:my_timer, microseconds:500))
    wait = wait + 500;
end if;

// stand-alone bridge assignment statement
bridge my_timer = TIM::timer_start(microseconds:500, event_inst:my_evt);

// State 4. "Raising"
needle_position = SPPIO::raise_needle (
    radial_position:self.radial_position,
    theta_offset:self.theta_offset,
    probe_id:self.probe_ID );
```

### 10.2.4 Avoiding Ambiguity in Invocations

Ambiguity among class and bridge operations can arise if the following conditions are present within a single domain:

- An external entity (EE) and a class share the same keyletters.
- The EE has a bridge operation with the same name as a class operation associated with the class.

When set, the Class Keyletters check-box under *Preferences | User | Audits* will report any EE's and Classes that share the same key letters. The key letter ambiguity should be removed by changing the key letter of either the EE or the class.

## 10.3 Function Invocation

A function is an operation that is global to the domain being modeled. Unlike class operations and bridge operations, a function's scope is at the domain level.

### 10.3.1 Syntax

As a function expression:

```
::<function name> (data item>:<expression>, ...)
```

As a stand-alone function assignment statement:

```
<variable> = ::<function name> (<data item>:<expression>, ...);
```

<function name> is the name of a function.

<data item> is the name of a data item input to <function name>.

<expression> is a string, arithmetic, boolean, simple or compound expression. The data type of the expression must match the data type defined for the given data item.

<variable> is a class attribute or a local variable.

### 10.3.2 Notes

It is strongly recommended that functions be named according to some convention that conveys their meaning throughout the domain.

A function expression may be used anywhere an expression is valid (e.g., assignment statement read values, control logic expressions, etc.).

The stand-alone function assignment statement must always be a stand-alone statement and cannot be used as an expression within another statement.



Since a function expression may be used anywhere an expression may be used, stand-alone function assignment statements are not necessary. The user may simply use a function expression as a read value and use an assign statement to assign the return value to <variable>.

Parentheses are required even if there are no data items.

If <variable> is a local variable that is being assigned for the first time, it will be of the same data type as the output value of <function name>.

If the type of <variable> has already been established (that is, if it is the attribute of a class or a local variable that has previously been assigned), then either:

- <variable> must be of the same data type as the output value of <function name>
- or
- the output value of <function name> is of type integer or real and <variable> is of type integer or real.

### 10.3.3 Example

```
// function expression without assigning the return value
::start();

// function expression as an assignment statement read value
cur_ext_time = ::current_external_time();

// function expression within the test part of an if statement
if ( ::shutdown() )
    generate S1:'Shutdown'() to S Creator;
end if;

// stand-alone function assignment statement
status = ::shutdown();
```

## 10.4 Object Action Language for Invocations

OAL can be specified for class and instance-based operations, bridge operations, and functions. The syntax rules applied to actions for these differ slightly from those applied to state actions. These differences are described in detail below.

### 10.4.1 Return Statement

Since class operations, bridge operations, and functions can return a value, the return statement is accepted within their actions.

#### Syntax

```
return <expression>;  
return;
```

<expression> is a string, arithmetic, boolean, simple or compound expression. The data type of the expression must match the data type defined for the return value of the class operation, bridge operation, or function.

#### Notes

When executed, the return statement causes control to be returned to the caller.

The value returned to the caller is <expression>.

If the return value of the class operation, bridge operation, or function is void, then <expression> must be omitted.

## Example

```
select any dog from instances of DOG;
return dog.weight;

// SSPIO: Bridge "Lower needle"
select any probe from instances of SP
  where selected.probe_ID == param.probe_id;
generate SP3:probe_in_position to probe;
return "down";
```

### 10.4.2 Parameters

Class and instance-based operations, bridge operations, and functions can accept parameters. These parameters can be accessed as read values by using the `param` keyword within the action. They can be modified if they are by-reference parameters.

## Syntax

```
param.<parameter>
```

<parameter> is the name of a parameter.

## Note

`param.<parameter>` is an <expression> and so can be used in any OAL construct specifying an expression.

## Example

```
// For an invocation like MATH::SQR(x:3)
// Return x**2
return param.x * param.x;
```

### 10.4.3 Use of self keyword

Instance-based operations can use the keyword `self` to reference the instance to which the operation is currently being applied. The `self` keyword can be used anywhere that is valid.

## Note

Class-based operations, bridge operations, and functions can not use the `self` keyword since they are not related to a specific instance.

## Example

```
// attribute access
self.a = self.b;
// event generation
generate E1:'one'() to self;
```

### 10.4.4 Other Differences

Class and instance-based operations, bridge operations, and functions may not refer to the `rcvd_evt` keyword. This keyword is only allowed within a state action since these are the only actions that can receive events.

---

## 11.1 External and Internal Time

This section describes statements that support date and time. The OAL supports two different concepts of time:

*External time:* Time as known in the external world. For example, 12 October 1492, 13:25:10. The accuracy of external time is dependent on the architecture and implementation.

*Internal time:* An internal system clock that measures time in “ticks”. The value of a tick is dependent upon the architecture and implementation.

### 11.1.1 External Time

#### Syntax

To create a <date variable>, write

```
[bridge] <date variable> = TIM::create_date (day:<arithmetic expression>,
month:<arithmetic expression>, year:<arithmetic expression>,
second:<arithmetic expression>, minute:<arithmetic expression>,
hour:<arithmetic expression>);
```

To read the current date or time, write

```
[bridge] <date variable> = TIM::current_date ();
```

To extract components of a <date variable>

```
<integer variable> = TIM::get_day (date:<date variable>);  
<integer variable> = TIM::get_month (date:<date variable>);  
<integer variable> = TIM::get_year (date:<date variable>);  
<integer variable> = TIM::get_second (date:<date variable>);  
<integer variable> = TIM::get_minute (date:<date variable>);  
<integer variable> = TIM::get_hour (date:<date variable>);
```

<arithmetic expression> is an arithmetic expression evaluating to an integer value.

<date variable> is a local variable or a class attribute of type date.

<integer variable> is a local variable or a class attribute of type integer.

## Note

External time is represented by a 24-hour clock.

### 11.1.2 Internal Time

## Syntax

To read the internal system clock, write

```
[bridge] <time variable> = TIM::current_clock ();
```

<time variable> is a local variable or a class attribute of type timestamp.

## Note

The system clock counts time in ticks. The size of a tick is dependent on the architecture and implementation.

# 12

## TIMERS

---

### 12.1 Starting a Timer

#### 12.1.1 Syntax

```
[bridge] <timer handle> = TIM::timer_start (microseconds:<arithmetic  
expression>, event_inst:<event instance>);
```

<timer handle> is a handle to a timer instance.

<arithmetic expression> is an expression that evaluates to an integer value.

<event instance> is a handle to an event instance.

This bridge operation starts a timer set to expire in <arithmetic expression> microseconds, generating the event <event instance> upon expiration. Returns the instance handle of the timer.

The bridge keyword is optional.

```
[bridge] <timer handle> = TIM::timer_start_recurring (microseconds:  
<arithmetic expression>, event_inst:<event instance>);
```

<timer handle> is a handle to a timer instance.

<arithmetic expression> is an expression that evaluates to an integer value.

<event instance> is a handle to an event instance.

This bridge operation starts a timer set to expire in <arithmetic expression> microseconds, generating the event <event instance> upon expiration. Upon expiration, the timer will be restarted and fire again in <arithmetic expression> microseconds generating the event <event instance>. This bridge operation returns the instance handle of the timer.

The bridge keyword is optional.

## 12.1.2 Example

```
// State 3. "Down"
select one row related by self->ROW[R2];
st = row.sampling_time;
create event instance move_on of SP1:finished_sampling() to self;
mo_timer = TIM::timer_start(microseconds:st, event_inst:move_on);
```

## 12.2 Querying a Timer

### 12.2.1 Syntax

```
[bridge] <integer variable> = TIM::timer_remaining_time (timer_inst_ref:
<timer handle>);
```

<integer variable> is a local variable or a class attribute of type integer.

<timer handle> is a handle to a timer instance.

Returns the time remaining (in microseconds) for the timer specified by <timer handle>. If the timer has expired, a zero value is returned.

The bridge keyword is optional.



## 12.3 Manipulating a Timer

### 12.3.1 Syntax

```
[bridge] <boolean variable> = TIM::timer_reset_time (timer_inst_ref:<timer handle>, microseconds:<arithmetic expression>);
```

<boolean variable> is a local variable or a class attribute of type boolean.

<timer handle> is a handle to a timer instance.

<arithmetic expression> is an expression that evaluates to an integer value.

This bridge operation attempts to set an existing timer <timer handle> to expire in <arithmetic expression> microseconds. If the timer exists (that is, it has not expired), a TRUE value is returned. If the timer no longer exists, a FALSE value is returned.

```
[bridge] <boolean variable> = TIM::timer_add_time (timer_inst_ref:<timer handle>, microseconds:<arithmetic expression>);
```

<boolean variable> is a local variable or a class attribute of type boolean.

<timer handle> is a handle to a timer instance.

<arithmetic expression> is an expression that evaluates to an integer value.

This bridge operation attempts to add <arithmetic expression> microseconds to an existing timer <timer handle>. If the timer exists (that is, it has not expired), a TRUE value is returned. If the timer no longer exists, a FALSE value is returned.

The bridge keyword is optional.

## 12.4 Canceling a Timer

### 12.4.1 Syntax

```
[bridge] <boolean variable> = TIM::timer_cancel (timer_inst_ref:<timer handle>);
```

<boolean variable> is a local variable or a class attribute of type boolean.

<timer handle> is a handle to a timer instance.

This bridge operation cancels and deletes the timer specified by <timer handle>. If the timer exists (that is, it had not expired), a TRUE value is returned. If the timer no longer exists, a FALSE value is returned.

The bridge keyword is optional.

### 12.4.2 Notes

When a timer fires, it is deleted unless it was created using the `timer_start_recurring` bridge operation.

In many architectures there may be a delay between the expiration of a timer and the delivery of the associated event to the receiving state machine.

# A

## REFERENCES

---

### A.1 References

The following published works were used in compiling this manual.

- Mel02 Stephen J. Mellor and Marc J. Balcer, *Executable UML: A Foundation for Model-Driven Architecture*, Addison-Wesley, Boston, MA, 2002
- Shl92 Sally Shlaer and Stephen J. Mellor, *Object Lifecycles: Modeling the World in States*, Prentice Hall, Englewood Cliffs, 1992
- Shl95 Sally Shlaer and Neil Lang, *Shlaer-Mellor Method: The OOA96 Report*, Project Technology, Inc., Berkeley, California, 1995
- Wil95 Ian Wilkie, Adrian King, and Mike Clarke, *The Action Specification Language (ASL) Reference Guide*, version 2.4, Kennedy-Carter, London, 1995



# B

## KEYWORDS

---

### B.1 Keywords

The following enumerates the list of reserved words.:

across	and	any	assign
assigner	break	bridge	by
cardinality	class	continue	create
creator	delete	each	elif
else	empty	end	event
false	for	from	generate
if	in	instance	instances
many	not	not_empty	object
of	one	or	param
rcvd_evt	relate	related	return
select	selected	self	to
transform	true	unrelate	using
where	while		

**Table B-1:** BridgePoint Object Action Language Keywords

Any keyword can appear in full upper case, full lower case or with the first character in upper case and the remaining characters in lower case.



---

## **C.1 Language Constructs**

### **C.1.1 Control Logic**

```
if (<boolean expression>
    // Executed if above boolean expression evaluates to TRUE
    <statements>
elif (<boolean expression>
    // Executed if above boolean expression evaluates to TRUE and previous
    boolean expression is FALSE
    <statements>
else
    // Executed if both boolean expressions evaluate to FALSE
    <statements>
end if;

for each <instance handle> in <instance handle set>
    <statements>
end for;

while <boolean expression>
    <statements>
end while;

break;

continue;
```

### **C.1.2 Instance Creation**

```
create object instance <instance handle> of <keyletter>;
```

```
create object instance of <keyletter>;
```

### **C.1.3 Instance Selection**

```
select any <instance handle> from instances of <keyletter> [ where <where  
expression> ];
```

```
select many <instance handle set> from instances of <keyletter> [ where  
<where expression> ];
```

### **C.1.4 Writing Attributes**

```
[assign] <instance handle>.<attribute> = <expression>;
```

```
[assign] self.<attribute> = <expression>; // Mathematically-dependent only
```

### **C.1.5 Reading Attributes**

```
[assign] <variable> = <instance handle>.<attribute>;
```

### **C.1.6 Instance Deletion**

```
delete object instance <instance handle>;
```



### **C.1.7 Creating Instances of a Relationship**

```
relate <source instance handle> to <destination instance handle> across  
  <relationship specification>;
```

```
relate <source instance handle> to <destination instance handle> across  
  <relationship specification> using <associative instance handle>;
```

### **C.1.8 Deleting Instances of a Relationship**

```
unrelate <source instance handle> from <destination instance handle> across  
  <relationship specification>;
```

```
unrelate <source instance handle> from <destination instance handle> across  
  <relationship specification> using <associative instance handle>;
```

### **C.1.9 Instance Selection by Relationship Navigation**

```
select one <instance handle> related by <start> -> <relationship link chain>  
  [ where <where expression> ];
```

```
select any <instance handle> related by <start> -> <relationship link chain>  
  [ where <where expression> ];
```

```
select many <instance handle set> related by <start> -> <relationship link  
  chain> [ where <where expression> ];
```

### **C.1.10 Creating Events**

```
create event instance <event instance> of <event label>[:<event meaning>]  
  [(<event parameters>)] to <target>;
```

### **C.1.11 Generating Events**

```
generate <event label>[:<event meaning>] [(<event parameters>)] to <target>;
```

```
generate <event instance>;
```

```
generate <instance handle>.<attribute>;
```

### **C.1.12 Accessing Event Data**

```
rcvd_evt.<supplemental data item>
```

### **C.1.13 Arithmetic, Logical, and String Assignment**

```
[assign] <boolean var> = <boolean expression>;
```

```
[assign] <arithmetic var> = <arithmetic expression>;
```

```
[assign] <string var> = <string expression>;
```

### **C.1.14 Unary Operators**

```
empty <handle>
```

```
not_empty <handle>
```

```
cardinality <handle>
```

### **C.1.15 Operations**

```
[transform] <keyletter>::operation name (<data item>:<expression>, ...);
```

```
[assign] <variable> = [transform] <keyletter>::operation name (<data item>:  
<expression>, ...);
```

### **C.1.16 Bridges**

```
[bridge] <eekeyletter>::bridge name (<data item>:<expression>, ...);
```

```
[assign] <variable> = [bridge] <eekeyletter>::bridge name (<data item>:  
<expression>, ...);
```

### **C.1.17 Functions**

```
::<function name> (<data item>:<expression>, ...);
```

```
[assign] <variable> = ::<function name> (<data item>:<expression>, ...);
```

### **C.1.18 Object Action Language for Non-State Actions**

```
return <expression>;
```

```
return;
```

```
param.<parameter>
```

## C.1.19 Date and Time

```
[bridge] <date variable> = TIM::create_date (day:<arithmetic expression>,
month:<arithmetic expression>, year:<arithmetic expression>, second:
<arithmetic expression>, minute:<arithmetic expression>, hour:<arithmetic
expression>);
```

```
[bridge] <date variable> = TIM::current_date ();
```

```
[bridge] <time variable> = TIM::current_clock ();
```

```
<integer variable> = TIM::get_day (date:<date variable>);
```

```
<integer variable> = TIM::get_month (date:<date variable>);
```

```
<integer variable> = TIM::get_year (date:<date variable>);
```

```
<integer variable> = TIM::get_second (date:<date variable>);
```

```
<integer variable> = TIM::get_minute (date:<date variable>);
```

```
<integer variable> = TIM::get_hour (date:<date variable>);
```

## C.1.20 Timers

```
[bridge] <timer handle> = TIM::timer_start (microseconds:<arithmetic
expression>, event_inst:<event instance>);
```

```
[bridge] <timer handle> = TIM::timer_start_recurring (microseconds:<arithmetic
expression>, event_inst:<event instance>);
```

```
[bridge] <integer variable> = TIM::timer_remaining_time (timer_inst_ref:<timer
handle>);
```

```
[bridge] <boolean variable> = TIM::timer_reset_time (timer_inst_ref:<timer
handle>, microseconds:<arithmetic expression>);
```

```
[bridge] <boolean variable> = TIM::timer_add_time (timer_inst_ref:<timer handle>, microseconds:<arithmetic expression>);
```

```
[bridge] <boolean variable> = TIM::timer_cancel (timer_inst_ref:<timer handle>);
```

## C.2 Statement Components

<arithmetic expression>	an expression evaluating to a real or an integer value. Used in places where only an integer is required.
<arithmetic operator>	a +, -, *, /, or % (remainder from arithmetic division)
<attribute>	the name of an attribute
<binary operator>	an and, or, +, -, *, /, or %
<boolean expression>	an expression evaluating to TRUE or FALSE
<bridge name>	the name of a bridge defined for the external entity specified by <eekeyletter>.
<eekeyletter>	the keyletter of an external entity
<event instance>	a local variable of type event instance
<event label>	the <keyletter><event number>
<event meaning>	the meaning of the event, as in 'turn off the light'; the tick marks may be omitted if the event meaning contains no spaces.
<event parameters>	the supplemental data items (if any) to be carried by the event. Each data item is given in the form <supplemental data item>:<expression>.
<handle>	an <instance handle>, <instance handle set>, or a timer handle (or a bridge, operation, or function invocation that returns a timer handle).
<instance handle>	a local variable referring to a single instance
<instance handle set>	a local variable referring to a set of instance handles
<keyletter>	the key letters of a class
<operation name>	the name of a operation defined for the class specified by <keyletter>.
<read value>	a readable value: a constant, local variable, <instance handle>.<attribute>, rcvd_evt.<supplemental data item>, param.<parameter>, or an invocation of an operation, bridge operation, or function.
<relationship link>	a <keyletter>[<relationship specification>], where the square brackets are literal and do not indicate optional text.

**Table C-1:** Statement Components

<relationship link chain>	<relationship link> or <relationship link> -> ... -> <relationship link>
<relationship phrase>	the text description of the relationship enclosed in tick marks
<relationship specification>	R<number> or R<number>.<relationship phrase>
<simple string>	any of the following that evaluates to a string value: string constant, local variable, attribute, operation invocation, bridge invocation, function invocation, or a supplemental data item received from an event.
<string expression>	an expression evaluating to a string value
<supplemental data item>	the name of a supplemental data item
<unary operator>	a unary operator such as not for boolean expressions and + and - for arithmetic expressions.
<unique id>	a variable of type unique ID (or an operation, bridge or function invocation that returns a unique ID).
<where expression>	a special boolean expression at the end of a select statement; it must contain the selected keyword.

**Table C-1:** Statement Components

## C.3 Production Rules

*This section contains the production rules for the language expressed in EBNF.*

**statement** ::= ( assignment\_statement | break\_statement | bridge\_statement | bridge\_or\_operation\_statement | continue\_statement | create\_event\_statement | create\_object\_statement | delete\_statement | empty\_statement | for\_statement | function\_statement | generate\_statement | if\_statement | operation\_statement | relate\_statement | return\_statement | select\_statement | unrelate\_statement | while\_statement ) SEMI ;

**assignment\_statement** ::= [ ASSIGN ] assignment\_expr ;

**break\_statement** ::= BREAK ;

**bridge\_statement** ::= BRIDGE [ ( local\_variable | attribute\_access ) EQUAL ] bridge\_invocation ;

**bridge\_or\_operation\_statement** ::= bridge\_or\_operation\_invocation ;

**continue\_statement** ::= CONTINUE ;

**create\_event\_statement** ::= CREATE EVENT INSTANCE local\_variable OF event\_spec ;

**create\_object\_statement** ::= CREATE OBJECT INSTANCE [ ( local\_variable OF )? local\_variable ] OF object\_keyletters ;

**delete\_statement** ::= DELETE OBJECT INSTANCE inst\_ref\_var ;

**empty\_statement** ::= ;

**for\_statement** ::= FOR EACH local\_variable IN inst\_ref\_set\_var ( statement )\* ( END FOR | Eof ) ;

**function\_statement** ::= [ ( local\_variable | attribute\_access ) EQUAL ] function\_invocation ;

**generate\_statement** ::= GENERATE ( event\_spec | local\_variable ) ;

**if\_statement** ::= IF expr ( statement )\* [ ( ELIF expr ( statement )\* )+ ] [ ELSE ( statement )\* ] ( END IF | Eof ) ;

**operation\_statement** ::= TRANSFORM [ ( local\_variable | attribute\_access ) EQUAL ] operation\_invocation ;

**relate\_statement** ::= RELATE inst\_ref\_var TO inst\_ref\_var ACROSS relationship [ DOT phrase ] [ USING assoc\_obj\_inst\_ref\_var ] ;

**return\_statement** ::= RETURN [ expr ] ;

**select\_statement** ::= SELECT ( ONE local\_variable object\_spec | ANY local\_variable object\_spec | MANY local\_variable object\_spec ) ;

**unrelate\_statement** ::= UNRELATE inst\_ref\_var FROM inst\_ref\_var ACROSS relationship [ DOT phrase ] [ USING assoc\_obj\_inst\_ref\_var ] ;

**while\_statement** ::= WHILE expr ( statement )\* ( END WHILE | Eof ) ;

**assignment\_expr** ::= ( local\_variable EQUAL )? local\_variable EQUAL expr | ( attribute\_access EQUAL )? attribute\_access EQUAL expr | event\_data\_access EQUAL expr ;

**attribute\_access** ::= inst\_ref\_var DOT attribute ;

**bridge\_invocation** ::= ee\_keyletters DOUBLECOLON bridge\_function LPAREN [ bridge\_or\_operation\_parameters ] RPAREN ;

**bridge\_or\_operation\_invocation** ::= obj\_or\_ee\_keyletters DOUBLECOLON function\_name LPAREN [ bridge\_or\_operation\_parameters ] RPAREN ;

**bridge\_or\_operation\_expr** ::= BRIDGE bridge\_invocation | TRANSFORM operation\_invocation | bridge\_or\_operation\_invocation ;



**bridge\_or\_operation\_parameters** ::= bridge\_or\_operation\_data\_item COLON expr ( COMMA  
bridge\_or\_operation\_data\_item COLON expr)\* ;

**event\_data\_access** ::= RCVD\_EVT DOT supp\_data\_item ;

**event\_spec** ::= event\_label [ COLON event\_meaning ] [ LPAREN [ supp\_data ] RPAREN ] TO ( ( ( object\_keyletters CLASS )? object\_keyletters CLASS | ( object\_keyletters CREATOR )? object\_keyletters CREATOR ) | ( inst\_ref\_var\_or\_ee\_keyletters ) ) ;

**function\_invocation** ::= DOUBLECOLON function\_function LPAREN [ function\_parameters ] RPAREN ;

**function\_parameters** ::= function\_data\_item COLON expr ( COMMA function\_data\_item COLON expr)\* ;

**inst\_ref\_var\_or\_ee\_keyletters** ::= ( local\_variable | GENERAL\_NAME | kw\_as\_id3 ) ;

**instance\_chain** ::= local\_variable ( ARROW object\_keyletters LSQBR relationship [ DOT phrase ] RSQBR )+ ;

**object\_spec** ::= ( RELATED BY instance\_chain | FROM INSTANCES OF object\_keyletters ) [ WHERE expr ] ;

**param\_data\_access** ::= PARAM DOT bridge\_or\_operation\_data\_item ;

**supp\_data** ::= supp\_data\_item COLON expr ( COMMA supp\_data\_item COLON expr)\* ;

**operation\_invocation** ::= object\_keyletters DOUBLECOLON operation\_function LPAREN [ bridge\_or\_operation\_parameters ] RPAREN ;

**where\_spec** ::= expr ;

**assoc\_obj\_inst\_ref\_var** ::= inst\_ref\_var ;

**attribute** ::= general\_name ;

**bridge\_function** ::= function\_name ;

**bridge\_or\_operation\_data\_item** ::= data\_item\_name ;

**data\_item\_name** ::= general\_name ;

**keyletters** ::= general\_name ;

**ee\_keyletters** ::= keyletters ;

**event\_label** ::= general\_name ;

**event\_meaning** ::= ( phrase | general\_name ) ;

**function\_data\_item** ::= data\_item\_name ;

**function\_function** ::= function\_name ;

**general\_name** ::= ( limited\_name | GENERAL\_NAME | kw\_as\_id2 | kw\_as\_id4 ) ;

**limited\_name** ::= ID | RELID ;

**inst\_ref\_set\_var** ::= local\_variable ;

**inst\_ref\_var** ::= local\_variable ;

**kw\_as\_id1** ::= ACROSS .. USING;

**kw\_as\_id2** ::= ACROSS .. TRUETOKEN;

**kw\_as\_id3** ::= BRIDGE .. TRUETOKEN;

**kw\_as\_id4** ::= PARAM .. SELF;

**local\_variable** ::= ( limited\_name | kw\_as\_id1 | SELECTED | SELF | INVALID\_CHARACTERS ) ;

**function\_name** ::= general\_name ;

**obj\_or\_ee\_keyletters** ::= keyletters ;

**object\_keyletters** ::= keyletters ;

**phrase** ::= ( PHRASE | BADPHRASE\_NL | Eof ) ;

**relationship** ::= RELID ;

**supp\_data\_item** ::= data\_item\_name ;

**operation\_function** ::= function\_name ;

**expr** ::= sub\_expr ;

**sub\_expr** ::= conjunction ( OR conjunction ) \* ;

**conjunction** ::= relational\_expr ( AND relational\_expr ) \* ;

**relational\_expr** ::= addition [ COMPARISON\_OPERATOR addition ] ;

**addition** ::= multiplication ( PLUS\_OR\_MINUS multiplication ) \* ;

**multiplication** ::= boolean\_negation | sign\_expr ( MULT\_OP sign\_expr ) \* ;

**sign\_expr** ::= [ PLUS | MINUS ] term ;

**boolean\_negation** ::= NOT term ;

**term** ::= ( CARDINALITY | EMPTY | NOTEMPTY ) local\_variable | rval | LPAREN ( ( assignment\_expr ) ? assignment\_expr | expr ) RPAREN ;

**rval** ::= constant\_value | variable | attribute\_access | event\_data\_access | bridge\_or\_operation\_expr |  
param\_data\_access | QMARK ;

**variable** ::= local\_variable ;

**constant\_value** ::= ( FRACTION | NUMBER | TRUETOKEN | FALSETOKEN ) | quoted\_string ;

**quoted\_string** ::= QUOTE ( STRING | BADSTRING\_NL | Eof ) ;



---

## **Symbols**

- 52  
!= 54  
% 52  
\* 52  
+ 52  
/ 52  
< 54  
<= 54  
== 54  
> 54  
>= 54

## **A**

**Across**  
in relate statement 36

**Action**  
semantics 1, 3  
types of 4

**And** 49, 50, 54

**Assign**  
and data typing 10  
for writing 28, 30  
variables 56

**Assigner** 44

## **Associative relationship**

creating 37  
unrelating 38

## **Attributes**

mathematically-dependent 28, 29  
reading 31  
writing 28

## **Autosampler** 6

## **B**

**Break** 21

**Bridge** 63, 66

## **C**

**Cardinality** 58

**Comments** 7

**Constants** 57

**Contacting Project Technology** 2

**Continue** 22

## **Create**

associative instance 37  
create\_date 71  
event instance 46  
object instance 25  
relationship 36

**Creator** 44

## **D**

### **Data items**

within an action 9

### **Data types** 10

declaration statements 10

handle set 11

instance handle 11

### **Delete**

object instance 32

relationship 33, 38

unconditional relationship 39

### **Domain**

data types specific to 10

external entities as 64

## **E**

### **Each** 19

### **EBNF** 89

### **Elif** 17

### **Else** 17

### **Empty**

function 58

handle set 11, 41

tick marks 45

### **End** 20

### **End if** 17

### **Event**

external entities, to 44

generation 44

example 45

instance 12, 46

supplemental data 43, 44, 45, 70

example 29

### **Example model** 6

## **Expressions**

arithmetic 52

boolean 53

compound 50

string 55

where expression 55

## **External entity** 63

and domains 64

events to 44

## **F**

### **Failure**

in relationship navigation 41

### **False** 53

### **For** 19

### **For Each** 19

### **From instances of** 26

## **G**

### **Generate** 47

See Events 44

## **H**

### **Handle**

See Instance handle and Instance handle set

## **I**

### **If** 17

### **Initialization**

of variables 13

**Instance** 25

- creating 25
- deleting 32

**Instance handle** 12

- creation of 25
- data type 11
- data type of 'self' 11
- defined 11
- empty in relationship navigation 41
- existence of 58
- in deleting an instance of a relationship 38
- in for each 19, 20
- in select 26
- relating 36
- self 25, 30
  - example usage 31, 32
  - within MDA actions 30

**Instance handle set** 12

- defined 11
- empty 11, 41
- empty in relationship navigation 41
- in for each 19
- in select 26
- size of 58

**L****Logical operation** 54**M****Mathematically-dependent attributes** 28, 29

- self keyword 30

**N****Naming rules** 8**Not** 53, 54**Not\_empty** 58**O****Of** 25**Or** 49, 50, 54**P****Param** 69**Phone numbers** 2**Production rules** 89**Project Technology**

- contacting 2
- website 2

**R****Rcvd\_evt**

- See Event supplemental data

**Read value** 13, 49, 50, 62, 63, 88

- example 63

**Relate** 36**Related by**

- in select statement 40

**Relationship**

- deletion of super/subtype 39
- phrase 35
- specification 35
- specifications 35
- unrelate 38

**Relationship navigation** 40**Return** 68

## **S**

### **Scope**

control logic structures 14  
definition 14

### **Select**

any 26, 40  
many 26, 40  
one 40  
related by 40

### **Selected**

use in where expressions 55

### **Self** 9, 11

data type of 11  
in relationship creation 37  
in relationship navigation 38  
use in create statement 25

### **Super/subtype relationships**

deletion of 39

### **Supplemental data item** 43

in generated events 44

### **Syntax specification** 89

## **T**

### **Tick marks** 35

### **Timer handle** 12

### **Transform** 61

### **True** 53

## **U**

### **Unconditional relationship**

creation of 25  
deletion of 39

### **Unrelate** 38

## **Using**

in relate statement 36

## **V**

### **Variable initialization** 13

## **W**

### **Where clause**

in select statement 26

### **While** 20

### **White space** 8